Chapter 42
INSET and INSET2 Statements

## Chapter Table of Contents

<table>
<thead>
<tr>
<th>Section</th>
<th>Page</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>OVERVIEW</strong></td>
<td>1715</td>
</tr>
<tr>
<td><strong>GETTING STARTED</strong></td>
<td>1716</td>
</tr>
<tr>
<td>Displaying Summary Statistics on a Control Chart</td>
<td>1716</td>
</tr>
<tr>
<td>Formatting Values and Customizing Labels</td>
<td>1718</td>
</tr>
<tr>
<td>Adding a Header and Positioning the Inset</td>
<td>1719</td>
</tr>
<tr>
<td><strong>SYNTAX</strong></td>
<td>1721</td>
</tr>
<tr>
<td>Summary of INSET Keywords</td>
<td>1722</td>
</tr>
<tr>
<td>Summary of Options</td>
<td>1724</td>
</tr>
<tr>
<td>Dictionary of Options</td>
<td>1725</td>
</tr>
<tr>
<td><strong>DETAILS</strong></td>
<td>1728</td>
</tr>
<tr>
<td>Positioning the Inset Using Compass Points</td>
<td>1728</td>
</tr>
<tr>
<td>Positioning the Inset in the Margins</td>
<td>1729</td>
</tr>
<tr>
<td>Positioning the Inset Using Coordinates</td>
<td>1729</td>
</tr>
</tbody>
</table>
Part 9. The CAPABILITY Procedure
Chapter 42
INSET and INSET2 Statements

Overview

The INSET and INSET2 statements allow you to enhance a Shewhart chart by adding a box or table (referred to as an inset) of summary statistics directly to the graph. The INSET statement places an inset in a primary Shewhart chart while the INSET2 statement places one in a secondary Shewhart chart. An inset can display statistics calculated by the SHEWHART procedure or arbitrary values provided in a SAS data set.

Note that an INSET or INSET2 statement by itself does not produce a display but must be used in conjunction with a chart statement. Insets are not available with line printer output, so the INSET and INSET2 statements are not applicable when the LINEPRINTER option is specified in the PROC SHEWHART statement.

You can use options in the INSET and INSET2 statements to

- specify the position of the inset
- specify a header for the inset table
- specify graphical enhancements, such as background colors, text colors, text height, text font, and drop shadows

The INSET2 statement differs from the INSET statement in only two respects.

1. An INSET2 statement creates an inset within a secondary chart generated by an IRCHART, MRCHART, XRCHART or XSCHART statement or by the TRENDVAR= option. For example, when following an XRCHART statement an INSET statement produces an inset in the $\bar{X}$ chart and an INSET2 statement produces one in the $R$ chart.

2. The INSET statement can be used to place an inset in one of the margins surrounding the plot area, while the INSET2 statement cannot.

Any of the statistics available for display in an inset can be specified with either an INSET or INSET2 statement. Descriptions of the INSET statement in this chapter also apply to the INSET2 statement except where explicitly noted.
Getting Started

This section introduces the INSET statement with examples that illustrate commonly used options. Complete syntax for the INSET statement is presented in the “Syntax” section on page 1721.

Displaying Summary Statistics on a Control Chart

In the manufacture of silicon wafers, batches of five wafers are sampled, and their diameters are measured in millimeters. The following statements create a SAS data set named WAFERS, which contains the measurements for 25 batches:

```sas
data wafers;
  input batch @;
  do i=1 to 5;
    input diamtr @;
    output;
  end;
  drop i;
cards;
  1 35.00 34.99 34.99 34.98 35.00
  2 35.01 34.99 34.99 34.98 35.00
  3 34.99 35.00 35.00 35.00 35.00
  4 35.01 35.00 34.99 34.99 35.00
  5 35.00 34.99 34.98 34.99 35.00
  6 34.99 34.99 35.00 35.00 35.00
  7 35.01 34.98 35.00 35.00 34.99
  8 35.00 35.00 34.99 34.98 34.99
  9 34.99 34.98 34.98 35.01 35.00
 10 34.99 35.00 35.01 34.99 35.01
 11 35.01 35.00 35.00 34.98 34.99
 12 34.99 34.99 35.00 34.98 35.01
 13 35.01 34.99 34.98 34.99 34.99
 14 35.00 35.00 34.99 35.01 34.99
 15 34.98 34.99 34.99 34.98 35.00
 16 34.99 35.00 35.00 35.01 35.00
 17 34.98 34.98 34.99 34.99 34.98
 18 35.01 35.02 35.00 34.98 35.00
 19 34.99 34.98 35.00 34.99 34.98
 20 34.99 35.00 35.00 34.99 34.99
 21 35.00 34.99 34.99 34.98 35.00
 22 35.00 35.00 35.01 35.00 35.00
 23 35.02 35.00 34.98 35.02 35.00
 24 35.00 35.00 34.99 35.01 34.98
 25 34.99 34.99 34.99 35.00 35.00
;
```

The following statements generate an $\bar{X}$ chart from the WAFERS data. Lower and upper specification limits for wafer diameters are given and the process capability index $C_p$ is computed. An INSET statement is used to display the specification limits,
the computed value of $C_p$ and the process standard deviation on the chart:

```sas
title 'Mean Chart with Inset';
proc shewhart data=wafers;
  xchart diamtr*batch /
    lsl = 34.97
    usl = 35.03;
  inset lsl usl cp stddev /
    height = 3;
run;
```

The resulting $\bar{X}$ chart is displayed in Figure 42.1. The INSET statement immediately follows the chart statement that creates the graphical display (in this case, the XCHART statement). Specify the keywords for inset statistics (such as LSL, USL, CP and STDDEV) immediately after the word INSET. The inset statistics appear in the order in which you specify the keywords. The HEIGHT= option on the INSET statement specifies the text height used to display the statistics in the inset.

A complete list of keywords that you can use with the INSET statement is provided in “Summary of INSET Keywords” on page 1722. Note that the set of keywords available for a particular display depends on both the plot statement that precedes the INSET statement and the options that you specify in the plot statement.

![Mean Chart for Diameters](chart.png)

**Figure 42.1.** An $\bar{X}$ Chart with an Inset

The following examples illustrate options commonly used for enhancing the appearance of an inset.
Formatting Values and Customizing Labels

By default, each inset statistic is identified with an appropriate label, and each numeric value is printed using an appropriate format. However, you may want to provide your own labels and formats. For example, in Figure 42.1 the default format used for $C_p$ and the process standard deviation prints an excessive number of decimal places. The following statements produce $\bar{X}$ and $R$ charts, each with its own inset. The unwanted decimal places are eliminated and the default specification limits labels are replaced with abbreviations:

```
proc shewhart data=wafers;
  xrchart diamtr*batch /
    lsl = 34.97
    usl = 35.03;
  inset lsl='LSL' usl='USL' /
    height = 3;
  inset2 cp (6.4) stddev (6.4) /
    height = 3;
run;
```

The resulting $\bar{X}$ and $R$ charts are displayed in Figure 42.2. You can provide your own label by specifying the keyword for that statistic followed by an equal sign (=) and the label in quotes. The label can have up to 24 characters.

The format 6.4 specified in parentheses after the CP and STDDEV keywords displays those statistics with a field width of six and four decimal places. In general, you can specify any numeric SAS format in parentheses after an inset keyword. You can also specify a format to be used for all the statistics in the INSET statement with the FORMAT= option (see the next example, “Adding a Header and Positioning the Inset”). For more information about SAS formats, refer to Chapter 14 of SAS Language: Reference, Version 6, First Edition.

Note that if you specify both a label and a format for a statistic, the label must appear before the format.
Adding a Header and Positioning the Inset

In the previous examples, the insets are displayed in the upper left corners of the plots, the default position for insets added to control charts. You can control the inset position with the POSITION= option. In addition, you can display a header at the top of the inset with the HEADER= option. The following statements create a data set to be used with the INSET DATA= keyword and the chart shown in Figure 42.3:

```sas
data location;
  length _LABEL_ $ 10 _VALUE_ $ 12;
  input _LABEL_ _VALUE_ &;
  cards;
  Plant   Santa Clara
  Line    1
  Shift   2
;)
```

```sas
proc shewhart data=wafers;
  xchart diamtr*batch /
    lsl = 34.97
    usl = 35.03;
  inset data= location lsl='LSL' usl='USL' cp (6.4) stddev (6.4) /
    position = rm
    cshadow = black
    header = 'Summary Statistics';
run;
```
Part 9. The CAPABILITY Procedure

The header (in this case, *Summary Statistics*) can be up to 40 characters. Note that a relatively long list of inset statistics is requested. Consequently, POSITION=RM is specified to position the inset in the right margin. For more information about positioning, see “Details” on page 1728. The CSHADOW= option is used to display a drop shadow on this inset. The *options*, such as HEADER=, POSITION= and CSHADOW= are specified after the slash (/) in the INSET statement. For more details on INSET statement options, see “Dictionary of Options” on page 1725.

Note that the contents of the data set LOCATION appear before other statistics in the inset. The position of the DATA= keyword in the keyword list determines the position of the data set’s contents in the inset.

![Mean Chart for Diameters](image)

*Figure 42.3. Adding a Header and Repositioning the Inset*
Syntax

The syntax for the INSET and INSET2 statements is as follows:

```
INSET  keyword-list < / options >;
INSET2 keyword-list < / options >;
```

You can use any number of INSET and INSET2 statements in the SHEWHART procedure. Each INSET or INSET2 statement produces a separate inset and must follow one of the chart statements. The inset appears on every panel (page) produced by the last chart statement preceding it. The statistics are displayed in the order in which they are specified. The following statements produce a boxplot with two insets and an $\bar{X}$ and $R$ chart with one inset in the $\bar{X}$ chart and one in the $R$ chart.

```
proc shewhart data=wafers;
   boxchart diamtr * batch / lsl=34.9 target=35 usl=35.1;
   inset lsl target usl;
   inset cp cpk cpm;
   xrchart diamtr*batch;
   inset nmin nmax nout;
   inset2 nlow2 nhigh2;
run;
```

The statistics displayed in an inset are computed for a specific process variable using observations for the current BY group. For example, in the following statements, there are two process variables (WEIGHT and DIAMETER) and a BY variable (LOCATION). If there are three different locations (levels of LOCATION), then a total of six $\bar{X}$ charts are produced. The statistics in each inset are computed for a particular variable and location. The labels in the inset are the same for each $\bar{X}$ chart.

```
proc shewhart data=axles;
   by location;
   xchart (weight diameter) * batch / tests=1 to 8;
   inset ntests 1 to 8;
run;
```

The components of the INSET and INSET2 statements are described as follows.

**keyword-list**

can include any of the *keywords* listed in “Summary of INSET Keywords” on page 1722. Some *keywords*, such as NTESTS and DATA=, require operands specified immediately after the *keyword*. Also, some inset statistics are available only if you request chart statements and options for which those statistics are calculated. For example,

- the NHIGH2, NLOW2, NTESTS2, LCL2 and UCL2 keywords are available only when a secondary chart is produced with the IRCHART, MRCHART, XARCHART or XSCHART statements.
- the NTESTS *keyword* requires the TESTS= option;
- the NTESTS2 *keyword* requires the TESTS2= option;
The CAPABILITY Procedure

- the capability index keywords such as CPK all require one or more of the LSL=, USL= and TARGET= options.

By default, inset statistics are identified with appropriate labels, and numeric values are printed using appropriate formats. However, you can provide customized labels and formats. You provide the customized label by specifying the keyword for that statistic followed by an equal sign (=) and the label in quotes. Labels can have up to 24 characters. You provide the numeric format in parentheses after the keyword. Note that if you specify both a label and a format for a statistic, the label must appear before the format. For an example, see “Formatting Values and Customizing Labels” on page 1718.

**options**

appear after the slash (/) and control the appearance of the inset. For example, the following INSET statement uses two appearance **options** (POSITION= and CTEXT=):

```
inset n nmin nmax / position=ne ctext=yellow;
```

The POSITION= option determines the location of the inset, and the CTEXT= option specifies the color of the text of the inset.

See “Summary of Options” on page 1724 for a list of all available **options**, and “Dictionary of Options” on page 1725 for detailed descriptions. Note the difference between keywords and options; keywords specify the information to be displayed in an inset, whereas options control the appearance of the inset.

### Summary of INSET Keywords

All keywords available with the SHEWHART procedure’s INSET and INSET2 statements request a single statistic in an inset, except for the NTESTS, NTESTS2 and DATA= keywords. The NTESTS and NTESTS2 keywords each require a list of indexes specifying the tests for special causes whose counts of positive results are to be displayed:

```
inset ntests 1 2 3 4;
inset ntests2 1 to 4;
```

For each of the requested tests, the number of positive results for the test is displayed in the inset. So if tests 1 through 4 are requested the results occupy four lines in the inset.

The DATA= keyword specifies a SAS data set containing (label, value) pairs to be displayed in an inset. The data set must contain the variables _LABEL_ and _VALUE_. _LABEL_ is a character variable whose values provide labels for inset entries. _VALUE_ can be character or numeric, and provides values displayed in the inset. The label and value from each observation in the DATA= data set occupy one line in the inset. Figure 42.3 shows an inset containing entries from a DATA= data set.
Table 42.1. Summary Statistics

<table>
<thead>
<tr>
<th>MEAN</th>
<th>estimated or specified process mean</th>
</tr>
</thead>
<tbody>
<tr>
<td>N</td>
<td>nominal subgroup size</td>
</tr>
<tr>
<td>NMIN</td>
<td>minimum subgroup size</td>
</tr>
<tr>
<td>NMAX</td>
<td>maximum subgroup size</td>
</tr>
<tr>
<td>NOUT</td>
<td>number of subgroups outside control limits on primary chart</td>
</tr>
<tr>
<td>NLOW</td>
<td>number of subgroups below lower control limit on primary chart</td>
</tr>
<tr>
<td>NHIGH</td>
<td>number of subgroups above upper control limit on primary chart</td>
</tr>
<tr>
<td>NTESTS</td>
<td>number of positive results of tests for special causes on primary chart</td>
</tr>
<tr>
<td>STDDEV</td>
<td>estimated or specified process standard deviation</td>
</tr>
<tr>
<td>DATA=</td>
<td>(label, value) pairs from SAS-data-set</td>
</tr>
</tbody>
</table>

Table 42.2. Secondary Chart Summary Statistics

<table>
<thead>
<tr>
<th>NOUT2</th>
<th>number of subgroups outside control limits on secondary chart</th>
</tr>
</thead>
<tbody>
<tr>
<td>NLOW2</td>
<td>number of subgroups below lower control limit on secondary chart</td>
</tr>
<tr>
<td>NHIGH2</td>
<td>number of subgroups above upper control limit on secondary chart</td>
</tr>
<tr>
<td>NTESTS2</td>
<td>number of positive results of tests for special causes on secondary chart</td>
</tr>
</tbody>
</table>

Table 42.3. Specification Limits

<table>
<thead>
<tr>
<th>LSL</th>
<th>lower specification limit</th>
</tr>
</thead>
<tbody>
<tr>
<td>USL</td>
<td>upper specification limit</td>
</tr>
<tr>
<td>TARGET</td>
<td>target value</td>
</tr>
</tbody>
</table>
Table 42.4. Capability Indices and Confidence Limits

<table>
<thead>
<tr>
<th>Symbol</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>CIALPHA</td>
<td>( \alpha ) value for computing capability index confidence limits</td>
</tr>
<tr>
<td>CP</td>
<td>capability index ( C_p )</td>
</tr>
<tr>
<td>CPLCL</td>
<td>lower confidence limit for ( C_p )</td>
</tr>
<tr>
<td>CPUCL</td>
<td>upper confidence limit for ( C_p )</td>
</tr>
<tr>
<td>CPK</td>
<td>capability index ( C_{pk} )</td>
</tr>
<tr>
<td>CPKLC</td>
<td>lower confidence limit for ( C_{pk} )</td>
</tr>
<tr>
<td>CPKUCL</td>
<td>upper confidence limit for ( C_{pk} )</td>
</tr>
<tr>
<td>CPL</td>
<td>capability index ( C_{pl} )</td>
</tr>
<tr>
<td>CPLCL</td>
<td>lower confidence limit for ( C_{pl} )</td>
</tr>
<tr>
<td>CPLUC</td>
<td>upper confidence limit for ( C_{pl} )</td>
</tr>
<tr>
<td>CPM</td>
<td>capability index ( C_{pm} )</td>
</tr>
<tr>
<td>CPMLCL</td>
<td>lower confidence limit for ( C_{pm} )</td>
</tr>
<tr>
<td>CPMUCL</td>
<td>upper confidence interval for ( C_{pm} )</td>
</tr>
<tr>
<td>CPU</td>
<td>capability index ( C_{pu} )</td>
</tr>
<tr>
<td>CPULCL</td>
<td>lower confidence limit for ( C_{pu} )</td>
</tr>
<tr>
<td>CPUUCL</td>
<td>upper confidence limit for ( C_{pu} )</td>
</tr>
</tbody>
</table>

Summary of Options

The following table lists the INSET and INSET2 statement options. For complete descriptions, see “Dictionary of Options,” which follows this section.

Table 42.5. INSET Options

<table>
<thead>
<tr>
<th>Symbol</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>CFILL=color</td>
<td>specifies color of inset background</td>
</tr>
<tr>
<td>CFILLH=color</td>
<td>specifies color of header background</td>
</tr>
<tr>
<td>CFILMLEX=color</td>
<td>specifies color of frame</td>
</tr>
<tr>
<td>CHEADER=color</td>
<td>specifies color of header text</td>
</tr>
<tr>
<td>CSHADOW=color</td>
<td>specifies color of drop shadow</td>
</tr>
<tr>
<td>CTEXT=color</td>
<td>specifies color of inset text</td>
</tr>
<tr>
<td>DATA</td>
<td>specifies data units for ( POSITION=(x, y) ) coordinates</td>
</tr>
<tr>
<td>FONT=font</td>
<td>specifies font of text</td>
</tr>
<tr>
<td>FORMAT=format</td>
<td>specifies format of values in inset</td>
</tr>
<tr>
<td>HEADER=’quoted string’</td>
<td>specifies header text</td>
</tr>
<tr>
<td>HEIGHT=value</td>
<td>specifies height of inset text</td>
</tr>
<tr>
<td>NOFRAME</td>
<td>suppresses frame around inset</td>
</tr>
<tr>
<td>POSITION=position</td>
<td>specifies position of inset</td>
</tr>
<tr>
<td>REFPOINT=BR</td>
<td>BL</td>
</tr>
</tbody>
</table>
Dictionary of Options

The following entries provide detailed descriptions of options for the INSET and INSET2 statements. Terms used in this section are illustrated in Figure 42.4.

**CFILL=color | BLANK**

specifies the color of the background (including the header background if you do not specify the CFILLH= option).

If you do not specify the CFILL= option, then by default, the background is empty. This means that items that overlap the inset (such as subgroup data points or control limits) show through the inset. If you specify any value for the CFILL= option, then overlapping items no longer show through the inset. Specify CFILL=BLANK to leave the background uncolored and also to prevent items from showing through the inset.

**CFILLH=color**

specifies the color of the header background. By default, if you do not specify a CFILLH= color, the CFILL= color is used.

**CFRAME=color**

specifies the color of the frame. By default, the frame is the same color as the axis of the plot.

**CHEADER=color**

specifies the color of the header text. By default, if you do not specify a CHEADER= color, the CTEXT= color is used.

**CSHADOW=color | CS=color**

specifies the color of the drop shadow. See Figure 42.3 on page 1720 for an example. By default, if you do not specify the CSHADOW= option, a drop shadow is not
displayed.

**CTEXT=color**

specifies the color of the text. By default, the inset text color is the same as the other text on the plot.

**DATA**

specifies that data coordinates are to be used in positioning the inset with the `POSITION=` option. The `DATA` option is available only when you specify `POSITION= (x, y)`, and it must be placed immediately after the coordinates `(x, y)`. For details, see the entry for the `POSITION=` option or “Positioning the Inset Using Coordinates” on page 1729. See Figure 42.7 on page 1730 for an example.

**FONT=font**

specifies the font of the text. By default, the font is SIMPLEX if the inset is located in the interior of the plot, and the font is the same as the other text displayed on the plot if the inset is located in the exterior of the plot.

**FORMAT=format**

specifies a format for all the values displayed in an inset. If you specify a format for a particular statistic, then this format overrides the format you specified with the `FORMAT=` option. See Figure 42.3 on page 1720 for an example.

**HEADER='string'**

specifies the header text. The string cannot exceed 40 characters. If you do not specify the `HEADER=` option, no header line appears in the inset.

**HEIGHT=value**

specifies the height of the text.

**NOFRAME**

suppresses the frame drawn around the text.

**POSITION=position**

**POS=position**

determines the position of the inset. The position can be a compass point keyword, a margin keyword, or a pair of coordinates `(x, y)`. You can specify coordinates in axis percent units or axis data units. For more information, see “Details” on page 1728. By default, `POSITION=NW`, which positions the inset in the upper left (northwest) corner of the display.

**REFPOINT=BR | BL | TR | TL**

**RP=BR | BL | TR | TL**

specifies the reference point for an inset that is positioned by a pair of coordinates with the `POSITION=` option. Use the `REFPOINT=` option with `POSITION=` coordinates. The `REFPOINT=` option specifies which corner of the inset frame you want positioned at coordinates `(x, y)`. The keywords BL, BR, TL, and TR represent bottom left, bottom right, top left, and top right, respectively. See Example 42.8 for an example. The default is `REFPOINT=BL`.

If you specify the position of the inset as a compass point or margin keyword, the
REFPOINT= option is ignored. For more information, see “Positioning the Inset Using Coordinates” on page 1729.
Details

This section provides details on three different methods of positioning the inset using the POSITION= option. With the POSITION= option, you can specify

- compass points
- keywords for margin positions
- coordinates in data units or percent axis units

Positioning the Inset Using Compass Points

You can specify the eight compass points N, NE, E, SE, S, SW, W, and NW as keywords for the POSITION= option. The following statements create the display in Figure 42.5, which demonstrates all eight compass positions. The default is NW.

```
proc shewhart data=wafers;
   xchart diamtr*batch / tests= 1 to 8;
   inset ntests 1 / height=3 cfill=blank header='NW' pos=nw;
   inset ntests 2 / height=3 cfill=blank header='N ' pos=n ;
   inset ntests 3 / height=3 cfill=blank header='NE' pos=ne;
   inset ntests 4 / height=3 cfill=blank header='E ' pos=e ;
   inset ntests 5 / height=3 cfill=blank header='SE' pos=se;
   inset ntests 6 / height=3 cfill=blank header='S ' pos=s ;
   inset ntests 7 / height=3 cfill=blank header='SW' pos=sw;
   inset ntests 8 / height=3 cfill=blank header='W ' pos=w ;
run;
```

![Mean Chart for Diameters](image)

Figure 42.5. Insets Positioned Using Compass Points
Positioning the Inset in the Margins

Using the INSET statement you can also position an inset in one of the four margins surrounding the plot area using the margin keywords LM, RM, TM, or BM, as illustrated in Figure 42.6. The INSET2 statement cannot be used to produce an inset in a margin.

![Figure 42.6. Positioning Insets in the Margins](image)

For an example of an inset placed in the right margin, see Example 42.3. Margin positions are recommended if a large number of statistics are listed in the INSET statement. If you attempt to display a lengthy inset in the interior of the plot, it is likely that the inset will collide with the data display.

Positioning the Inset Using Coordinates

You can also specify the position of the inset with coordinates: POSITION= \((x, y)\). The coordinates can be given in axis percent units (the default) or in axis data units.

**Data Unit Coordinates**

If you specify the DATA option immediately following the coordinates, the inset is positioned using axis data units. For example, the following statements place the bottom left corner of the inset at 6 on the horizontal axis and 34.985 on the vertical axis:

```sas
proc shewhart data=wafers;
  xchart diamtr*batch;
  inset n / header = 'Position=(6,34.985)' position = (6,34.985) data;
run;
```
The control chart is displayed in Figure 42.7. By default, the specified coordinates determine the position of the bottom left corner of the inset. You can change this reference point with the REFPOINT= option, as in the next example.

**Figure 42.7.** Inset Positioned Using Data Unit Coordinates

### Axis Percent Unit Coordinates

If you do not use the DATA option, the inset is positioned using axis percent units. The coordinates of the bottom left corner of the display are (0, 0), while the upper right corner is (100, 100). For example, the following statements create a $\bar{X}$ chart with two insets, both positioned using coordinates in axis percent units:

```sas
proc shewhart data=wafers;
  xchart diamtr*batch;
inset nmin / position = (5,25)
    header = 'Position=(5,25)'
    height = 3
    refpoint = tl;
inset nmax / position = (95,95)
    header = 'Position=(95,95)'
    height = 3
    refpoint = tr;
run;
```

The display is shown in Figure 42.8. Notice that the REFPOINT= option is used to determine which corner of the inset is to be placed at the coordinates specified with the POSITION= option. The first inset has REFPOINT=TL, so the top left corner of the inset is positioned 5% of the way across the horizontal axis and 25% of the way up the vertical axis. The second inset has REFPOINT=TR, so the top right corner of the inset is positioned 95% of the way across the horizontal axis and 95% of the way up the vertical axis.
up the vertical axis. Note also that coordinates in axis percent units must be between 0 and 100.

![Mean Chart for Diameters](image)

**Figure 42.8.** Inset Positioned Using Axis Percent Unit Coordinates